
1. [bookmark: _Toc122434485][bookmark: _Toc52895930]Overview
This document lists all the changes needed to correct issues in the ATS iwd-TTCN3-B2022-09_D22wk49 related to the title of this CR.
Contact:	Kalahasti, Narendra
	Narendra.Kalahasti@anritsu.com

1. [bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc54186268]Corrections required
2.1 Change 1
	Function name
	f_TC_8_1_2_15_EUTRA

	Reason for change
	For CAT-M1 UEs, schedulingInfoList_BR_r13 IE should be consistent with schedulingInfoList according to section 6.2.2 of TS 36.331.
“Indicates additional scheduling information of SI messages for BL UEs and UEs in CE. It includes the same number of entries, and listed in the same order, as in schedulingInfoList (without suffix).”
But this is not taken care in the current TTCN implementation.

	Summary of change
	Configured missing third element on schedulingInfoList_BR_r13 IE for the corresponding SIB mapping info sibType29_v1610.

	TTCN module
	LTE\8_1\RRC_ConnEst.ttcn

	MCC160 Comment
	



Before change:
function f_TC_8_1_2_15_EUTRA() runs on EUTRA_PTC
  { /*  RRC connection establishment / Extended and spare fields in SI */
    /* @sic R5-206433 R5-206439 R5-206054 sic@ */
    var NAS_MSG_Indication_Type v_NasInd;
    var AccessStratumRelease v_Rel;
    var template (value) EUTRA_CellInfo_Type v_CellInfo;
    var SysinfoType_Type v_SysinfoType;
    var template (value) AllSiSchedul_Type v_SiSchedul;
    var SIB_Type v_SIB1_SibType1 := sibType29_v1610;    // by default value given for a UE Rel-8 to 15
    // Variables for case UE Rel-16
    var template(value) SystemInformationBlockType1_v12j0_IEs v_SystemInformationBlockType1_v12j0;

    f_EUTRA_Init(c1);
                                               
    //Create and configure cells
    f_EUTRA_CellConfig_Def(eutra_Cell1);

    // Preamble: The UE is in state Switched OFF (state 1)
    f_EUTRA_Preamble(eutra_Cell1, STATE2_IDLEUPDATE);

    //check the release: Rel-8 to Rel-16 only are allowed otherwise inconclusive verdict.
    v_Rel := f_EUTRA_MobileInfo_GetUECapability_AccessStratumRelease();
    if (not match (v_Rel, (rel8, rel9, rel10, rel11, rel13, rel14, rel15, rel16)))// NOTE: Spare field testing is not applicable for Rel-12 UE because a spare field in MIB or SIB1 does not exist in Rel-12.
    {
      f_SetVerdictInconc(__FILE__, __LINE__, "Test Case applicable to rel8, rel9, rel10, rel11, rel13, rel14, rel15, rel16 UE only");
    }
    
    f_EUTRA_UE_Detach_SwitchOffUe(eutra_Cell1, RRC_IDLE);

    f_EUTRA_TestBody_Set(true);
    
    //@siclog "Step 1" siclog@
    // System information that includes extended and spare fields that the UE does not comprehend is broadcasted on BCCH and BCCH-BR.
    v_CellInfo := f_EUTRA_CellInfo_Get(eutra_Cell1);
    v_SysinfoType := f_EUTRA_CellInfo_GetSysinfoType(eutra_Cell1);
    
    // Set MIB as per Table 8.1.2.15.3.3-1: Message MasterInformationBlock
    v_CellInfo.Sysinfo.BCCH_Info.MIB.message_.spare := '1'B;    /* @sic R5s221178 BASELINE MOVING 2022 rel-17 sic@ */
    v_CellInfo.Sysinfo.BCCH_Info.MIB.message_.partEARFCN_17.spare := '11'B;


    // Table 8.1.2.15.3.3-2: Message SystemInformationBlockType1
    if (v_Rel == rel16)
    { // test UE Rel-16
       //set schedulingInfoListExt-r12 in SIB1 with sibType32-v1700 for Message SystemInformationBlockType1 and SystemInformationBlockType1-BR-r13
       v_SystemInformationBlockType1_v12j0 := cs_SystemInformationBlockType1_v12j0(-, cs_SchedulingInfoListExt_1SI(rf32,{sibType32_v1700}));
       f_EUTRA_CellInfo_SetSysInfo_SIB1_v12j0(eutra_Cell1, v_SystemInformationBlockType1_v12j0);
    }
    else
    {   // UE is Rel-8 to Rel-15
        // Set scheduling in SIB1
       v_CellInfo.Sysinfo.BCCH_Info.SIB1.message_.c1.systemInformationBlockType1.schedulingInfoList[2].si_Periodicity := rf32;
       v_CellInfo.Sysinfo.BCCH_Info.SIB1.message_.c1.systemInformationBlockType1.schedulingInfoList[2].sib_MappingInfo[0] := v_SIB1_SibType1;
        
       // Set scheduling in SIB1-BR as per Table 8.1.2.15.3.3-3: Message SystemInformationBlockType1-BR-r13
       if (v_SysinfoType == sysinfoAndBR) {//UE supports category M1 or enhanced coverage TC
           v_CellInfo.Sysinfo.BCCH_BR_Info.SIB1.message_.c1.systemInformationBlockType1_BR_r13.schedulingInfoList[2].si_Periodicity := rf32;
           v_CellInfo.Sysinfo.BCCH_BR_Info.SIB1.message_.c1.systemInformationBlockType1_BR_r13.schedulingInfoList[2].sib_MappingInfo[0] := v_SIB1_SibType1;
       }
    }

    f_EUTRA_CellInfo_Set(eutra_Cell1, v_CellInfo);
    // Remove SIB26 from the scheduling list for SS configuration
    v_SiSchedul := f_EUTRA_SS_SetupSchedulingInfo(valueof(v_CellInfo.Sysinfo.BCCH_Info.SIB1.message_.c1.systemInformationBlockType1), f_EUTRA_CellInfo_GetFDD_TDD(eutra_Cell1), v_SysinfoType, valueof(v_CellInfo.Frequency.DL_ChBandwidth));
    v_SiSchedul.SiList := {v_SiSchedul.SiList[0], v_SiSchedul.SiList[1]};
    f_EUTRA_SS_ConfigureSysinfo(eutra_Cell1, -, -, -, -, v_SiSchedul);

    //@siclog "Step 2" siclog@
    //The UE is switched on.
    f_EUTRA_SwitchOnUEandStartIP(eutra_Cell1);

    //@siclog "Step 3" siclog@
    //Does the UE transmit an RRCConnectionRequest message on Cell 1?
    f_EUTRA_RRC_ConnectionRequest_Def(eutra_Cell1);
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 8.1.2.15 Step 3");

    //@siclog "Step 4" siclog@
    //The SS transmits an RRCConnectionSetup message on Cell 1.
    f_EUTRA_RRC_ConnectionSetup_Def ( eutra_Cell1);

    //@siclog "Step 5" siclog@
    //Check: Does the UE transmit an RRCConnectionSetupComplete message to confirm the successful completion of the connection establishment and to initiate the Attach procedure by including the ATTACH REQUEST message? The PDN CONNECTIVITY REQUEST message is piggybacked in ATTACH REQUEST.
    v_NasInd := f_EUTRA_RRCConnectionSetupComplete_Def(eutra_Cell1,
                                                       cr_508_RRCConnectionSetupComplete(tsc_RRC_TI_Def, ?),
                                                       cr_NAS_IndicationWithPiggybacking(tsc_SHT_IntegrityProtected,
                                                                                         cr_508_ATTACH_REQUEST(f_GetEAttachType(NORMAL),
                                                                                                               f_GetAdditionalUpdateType()),
                                                                                         cr_508_PDN_CONNECTIVITY_REQUEST(tsc_PdnRequest_InitialAttach, f_GetPdnType())));
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 8.1.2.15 Step 5");
    
    //@siclog "Step 6-18" siclog@
    //Steps 5 to 17 of the registration procedure described in TS 36.508 subclause 4.5.2.3 are performed on Cell 1.
    f_EUTRA_InitialRegistration_Step5_18b1 (eutra_Cell1, STATE2_IDLEUPDATE, NORMAL, v_NasInd);
    
    f_EUTRA_TestBody_Set(false);

    /* Postamble */
    f_EUTRA_Postamble(eutra_Cell1, E1_IDLE);
  }


After change:

import from EUTRA_SysInfo_Templates_BR all; 

function f_TC_8_1_2_15_EUTRA() runs on EUTRA_PTC
  { /*  RRC connection establishment / Extended and spare fields in SI */
    /* @sic R5-206433 R5-206439 R5-206054 sic@ */
    var NAS_MSG_Indication_Type v_NasInd;
    var AccessStratumRelease v_Rel;
    var template (value) EUTRA_CellInfo_Type v_CellInfo;
    var SysinfoType_Type v_SysinfoType;
    var template (value) AllSiSchedul_Type v_SiSchedul;
    var SIB_Type v_SIB1_SibType1 := sibType29_v1610;    // by default value given for a UE Rel-8 to 15
    // Variables for case UE Rel-16
    var template(value) SystemInformationBlockType1_v12j0_IEs v_SystemInformationBlockType1_v12j0;
    var BR_Si_Narrowband_Type v_BR_Si_Narrowband;

    f_EUTRA_Init(c1);
                                               
    //Create and configure cells
    f_EUTRA_CellConfig_Def(eutra_Cell1);

    // Preamble: The UE is in state Switched OFF (state 1)
    f_EUTRA_Preamble(eutra_Cell1, STATE2_IDLEUPDATE);

    //check the release: Rel-8 to Rel-16 only are allowed otherwise inconclusive verdict.
    v_Rel := f_EUTRA_MobileInfo_GetUECapability_AccessStratumRelease();
    if (not match (v_Rel, (rel8, rel9, rel10, rel11, rel13, rel14, rel15, rel16)))// NOTE: Spare field testing is not applicable for Rel-12 UE because a spare field in MIB or SIB1 does not exist in Rel-12.
    {
      f_SetVerdictInconc(__FILE__, __LINE__, "Test Case applicable to rel8, rel9, rel10, rel11, rel13, rel14, rel15, rel16 UE only");
    }
    
    f_EUTRA_UE_Detach_SwitchOffUe(eutra_Cell1, RRC_IDLE);

    f_EUTRA_TestBody_Set(true);
    
    //@siclog "Step 1" siclog@
    // System information that includes extended and spare fields that the UE does not comprehend is broadcasted on BCCH and BCCH-BR.
    v_CellInfo := f_EUTRA_CellInfo_Get(eutra_Cell1);
    v_SysinfoType := f_EUTRA_CellInfo_GetSysinfoType(eutra_Cell1);
    
    // Set MIB as per Table 8.1.2.15.3.3-1: Message MasterInformationBlock
    v_CellInfo.Sysinfo.BCCH_Info.MIB.message_.spare := '1'B;    /* @sic R5s221178 BASELINE MOVING 2022 rel-17 sic@ */
    v_CellInfo.Sysinfo.BCCH_Info.MIB.message_.partEARFCN_17.spare := '11'B;


    // Table 8.1.2.15.3.3-2: Message SystemInformationBlockType1
    if (v_Rel == rel16)
    { // test UE Rel-16
       //set schedulingInfoListExt-r12 in SIB1 with sibType32-v1700 for Message SystemInformationBlockType1 and SystemInformationBlockType1-BR-r13
       v_SystemInformationBlockType1_v12j0 := cs_SystemInformationBlockType1_v12j0(-, cs_SchedulingInfoListExt_1SI(rf32,{sibType32_v1700}));
       f_EUTRA_CellInfo_SetSysInfo_SIB1_v12j0(eutra_Cell1, v_SystemInformationBlockType1_v12j0);
    }
    else
    {   // UE is Rel-8 to Rel-15
        // Set scheduling in SIB1
       v_CellInfo.Sysinfo.BCCH_Info.SIB1.message_.c1.systemInformationBlockType1.schedulingInfoList[2].si_Periodicity := rf32;
       v_CellInfo.Sysinfo.BCCH_Info.SIB1.message_.c1.systemInformationBlockType1.schedulingInfoList[2].sib_MappingInfo[0] := v_SIB1_SibType1;
        
       // Set scheduling in SIB1-BR as per Table 8.1.2.15.3.3-3: Message SystemInformationBlockType1-BR-r13
       if (v_SysinfoType == sysinfoAndBR) {//UE supports category M1 or enhanced coverage TC
           v_BR_Si_Narrowband := f_EUTRA_InitBR_SiNarrowband(f_EUTRA_CellInfo_GetDL_ChBandwidth(eutra_Cell1));
           v_CellInfo.Sysinfo.BCCH_BR_Info.SIB1.message_.c1.systemInformationBlockType1_BR_r13.schedulingInfoList[2].si_Periodicity := rf32;
           v_CellInfo.Sysinfo.BCCH_BR_Info.SIB1.message_.c1.systemInformationBlockType1_BR_r13.schedulingInfoList[2].sib_MappingInfo[0] := v_SIB1_SibType1;
           v_CellInfo.Sysinfo.BCCH_BR_Info.SIB1.message_.c1.systemInformationBlockType1_BR_r13.nonCriticalExtension.nonCriticalExtension.nonCriticalExtension.nonCriticalExtension.nonCriticalExtension.bandwidthReducedAccessRelatedInfo_r13.schedulingInfoList_BR_r13[2] := cs_SchedulingInfo_BR(v_BR_Si_Narrowband);
       }
    }

    f_EUTRA_CellInfo_Set(eutra_Cell1, v_CellInfo);
    // Remove SIB26 from the scheduling list for SS configuration
    v_SiSchedul := f_EUTRA_SS_SetupSchedulingInfo(valueof(v_CellInfo.Sysinfo.BCCH_Info.SIB1.message_.c1.systemInformationBlockType1), f_EUTRA_CellInfo_GetFDD_TDD(eutra_Cell1), v_SysinfoType, valueof(v_CellInfo.Frequency.DL_ChBandwidth));
    v_SiSchedul.SiList := {v_SiSchedul.SiList[0], v_SiSchedul.SiList[1]};
    f_EUTRA_SS_ConfigureSysinfo(eutra_Cell1, -, -, -, -, v_SiSchedul);

    //@siclog "Step 2" siclog@
    //The UE is switched on.
    f_EUTRA_SwitchOnUEandStartIP(eutra_Cell1);

    //@siclog "Step 3" siclog@
    //Does the UE transmit an RRCConnectionRequest message on Cell 1?
    f_EUTRA_RRC_ConnectionRequest_Def(eutra_Cell1);
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 8.1.2.15 Step 3");

    //@siclog "Step 4" siclog@
    //The SS transmits an RRCConnectionSetup message on Cell 1.
    f_EUTRA_RRC_ConnectionSetup_Def ( eutra_Cell1);

    //@siclog "Step 5" siclog@
    //Check: Does the UE transmit an RRCConnectionSetupComplete message to confirm the successful completion of the connection establishment and to initiate the Attach procedure by including the ATTACH REQUEST message? The PDN CONNECTIVITY REQUEST message is piggybacked in ATTACH REQUEST.
    v_NasInd := f_EUTRA_RRCConnectionSetupComplete_Def(eutra_Cell1,
                                                       cr_508_RRCConnectionSetupComplete(tsc_RRC_TI_Def, ?),
                                                       cr_NAS_IndicationWithPiggybacking(tsc_SHT_IntegrityProtected,
                                                                                         cr_508_ATTACH_REQUEST(f_GetEAttachType(NORMAL),
                                                                                                               f_GetAdditionalUpdateType()),
                                                                                         cr_508_PDN_CONNECTIVITY_REQUEST(tsc_PdnRequest_InitialAttach, f_GetPdnType())));
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 8.1.2.15 Step 5");
    
    //@siclog "Step 6-18" siclog@
    //Steps 5 to 17 of the registration procedure described in TS 36.508 subclause 4.5.2.3 are performed on Cell 1.
    f_EUTRA_InitialRegistration_Step5_18b1 (eutra_Cell1, STATE2_IDLEUPDATE, NORMAL, v_NasInd);
    
    f_EUTRA_TestBody_Set(false);

    /* Postamble */
    f_EUTRA_Postamble(eutra_Cell1, E1_IDLE);
  }

